Noname manuscript No.
(will be inserted by the editor)

An Isabelle/HOL Formalisation of Green's Theorem

Mohammad Abdulaziz · Lawrence C. Paulson

the date of receipt and acceptance should be inserted later

Abstract We mechanise a proof of Green's theorem in Isabelle/HOL. We use a novel proof that avoids the ubiquitous line integral cancellation argument. This eliminates the need to formalise orientations and region boundaries explicitly with respect to the outwards-pointing normal vector. Instead we appeal to a homological argument about equivalences between paths. Contributions include mechanised theories of line integrals and partial derivatives, as well as the first mechanisation of Green's theorem.

1 Introduction

The *Fundamental Theorem of Calculus* (FTC) is a theorem of immense importance in differential calculus and its applications, relating a function's derivative to its integral. Having been conceived in the seventeenth century in parallel with the development of the infinitesimal calculus, more general forms of the FTC have been developed, the most general of which is the *General Stokes Theorem*.

A generalisation of the FTC (and a special case of the General Stokes Theorem in $\mathbb{R}^2$) was published in 1828 by George Green [4], with applications to electromagnetism in mind. Green's Theorem is the main topic of this work. In modern terms, it can be stated as follows:

Theorem 1. *Given a region $D \subseteq \mathbb{R}^2$ with an "appropriate" positively oriented boundary ∂D, and a field "appropriately" defined on D as $F(a) = (F_x(a), F_y(a))$,*

Mohammad Abdulaziz
Technical University of Munich
E-mail: mohammad.abdulaziz@in.tum.de

Lawrence Paulson
Computer Laboratory, University of Cambridge, England
E-mail: lp15@cam.ac.uk

for every $a \in D$, the following identity holds:

$$\int_D \frac{\partial F_y}{\partial x} - \frac{\partial F_x}{\partial y}\, dxdy = \oint_{\partial D} F_x dx + F_y dy,$$

where the left hand side is a double integral and the right hand side is a line integral in $\mathbb{R}^2$.

The term under double integral can be interpreted as the curl of the field F in the plane which, for instance, physically represents the vorticity of a physical field. The line integral of the field is on the region's boundary, and it can be physically interpreted as the work done by the field F on a particle moving along the path ∂D in the plane, for instance. Thus, this statement is a special case of the 3-dimensional Kelvin-Stokes theorem [2, p. 438]. Also note that one can obtain the 2-dimensional divergence theorem for the field F and a region D if the statement above is applied to $(-F_x(a), F_y(a))$ instead of $(F_x(a), F_y(a))$. This is because in that case the line integral will be $\oint_{\partial D} F_y dy - F_x dx$, which is the flux of F through ∂D, and the double integral will be $\int_D \frac{\partial F_x}{\partial y} + \frac{\partial F_y}{\partial x}\, dxdy$, which is the divergence of F through D.

Many statements of Green's theorem define, with varying degrees of generality,

- *the geometrical assumptions*: what is an appropriate boundary
- *the analytic assumptions*: what is an appropriate field.

The prevalent textbook form of Green's theorem asserts that, geometrically, the region can be divided into *elementary regions* and that, analytically, the field has smooth partial derivatives throughout the region. Also, the underlying integral is a Riemann integral in most textbooks.

Despite this being enough for most applications, more general forms of the theorem have appeared in the analysis literature. Michael [10] proves a statement of the theorem that generalises the geometrical assumptions, only assuming that the region has a rectifiable boundary (i.e. a boundary with finite length). Jurkat et al. [9] prove a statement of the theorem with exceptionally general analytic assumptions: they only assume that the field is continuous in the region, and that the *total* derivative of the field exists in the region except for a σ_1-finite set of points in the region. Then, they use that statement of Green's theorem to derive a general form of Cauchy's integral theorem.

Green's theorem has innumerable applications. In physics, these include electrodynamics and mechanics; in engineering these include deriving moments of inertia, hydrodynamics and the basis of the planimeter. Furthermore, Green's theorem is a fundamental result for a multitude of branches in mathematics, e.g. it can be used to derive Cauchy's integral theorem, and to justify efficient numerical solution methods for partial differential equations describing dynamical systems.

We formalise a statement of Green's theorem for Henstock-Kurzweil gauge integrals in the interactive theorem prover Isabelle/HOL [11]. Our work builds on the work of Hölzl et al. [7] and the Isabelle/HOL analysis library [8].

Existing proofs of Green's theorem all have one fundamental argument in common: showing that if a region is divided into subregions, the line integral of a field on the region's boundary is equal to the sum of the line integrals of the field around the subregions' boundaries. Crucially, line integrals on boundaries of neighbouring subregions cancel each other out. Formalising this argument depends on formalising orientations and region boundaries explicitly using for instance, an outward-pointing vector as used in Federer [3]. This can be hard, especially for regions with holes.

We have developed a novel proof. Avoiding the usual line cancellation argument, we use a homological argument that characterises equivalences between paths. Accordingly our formalisation does not strictly follow any published proof, but for reference we used the proofs by Zorich and Cooke [17], Spivak [14], and Protter [12].

Contributions

This paper extends our prior work [1], which presented a mechanisation of line integrals, partial derivatives and Green's theorem. One major difference between this paper and the conference version is that we provide a more general formalisation that removes all symmetric definitions and proofs. In particular, in our original formalisation we had separate definitions, theorems and proofs that were (almost) symmetric: they were once stated for the x-axis, and a second time for the y-axis with some modifications to accommodate the skew symmetry between the two axes. In the current formalisation, symmetric objects are replaced by objects parameterised by arbitrary orthonormal bases, thus eliminating most symmetries. A key component which made this generalisation possible was porting the multivariate change of variables theorem from HOL Light to Isabelle, which is a substantial formalisation in its own right (around 12K lines of proof script).

Another major difference is that we apply our statement of Green's theorem to simple regions (a diamond and a disk) as a tutorial showing how to use the theorem that we formalised. Moreover, we have substantially generalised the theorem statement as follows to simplify the treatment of applications.

- We have weakened the geometric assumptions of our theorem in two ways:
  - (i) generalising our definition of the equivalence of two paths from requiring that they have equal parameterisations to requiring that one of them is the reparameterisation of the other through a piecewise-smooth map ϕ
  - (ii) generalising our definition of 1-chains having a common sub-division to exclude a finite number of points from each 1-chain
- We have generalised our definition of elementary regions to have piecewise-smooth edges instead of smooth edges.
- And we have generalised our formalisation to accept parameterisations of regions that are either clockwise or anticlockwise orientations, versus our original restriction to only anticlockwise parameterisations.

We also elaborate more on the intuitions behind the concepts related to Green's theorem and our proof.

2 Isabelle Prerequisites

Before we proceed with explaining our formalisation, we first give a brief overview of some basic features of Isabelle.

2.1 Axiomatic Type Classes

Axiomatic type classes [16] are a powerful refinement of polymorphism, supporting principled overloading of notation and inheritance of common structures and their properties.

A *type class* denotes a collection of types; a *sort* is a list of type classes and denotes their intersection. Each type variable has a sort and can be instantiated by any type that belongs to all of the listed type classes. An *axiomatic type class* is a type class augmented with axioms constraining the constants. We can refer to these axioms in proofs, obtaining theorems specific to the type class. To show that a type τ is an instance of a particular axiomatic type class, we verify the corresponding axioms. These typically refer to overloaded constants, which we define for type τ with the objective of satisfying the axioms. Verifying the axioms for type τ makes all the theorems proved for the type class immediately available for type τ.

Isabelle/HOL uses type classes to organise the various numeric types (integers, rationals, complex numbers, etc.) and also a variety of topological concepts, such as metric spaces, topological spaces of various kinds and Euclidean spaces [8]. Most of the operators shown in the sequel, from the humble arithmetic operators to the various differentiation and integration operators, are overloaded through type classes.

2.2 The Isabelle/HOL Analysis Library

Isabelle/HOL is distributed with a comprehensive library covering topology, Euclidean spaces, complex analysis and much other material, mostly ported from the HOL Light multivariate analysis library [5,6]. Below we briefly introduce a few of the mathematical concepts used in the rest of the paper.

- In Isabelle/HOL Euclidean spaces are formalised as a type class. Most notably, it fixes the operators *norm* and $\cdot$ for the norm of a vector and the inner product of two vectors, respectively.
- A *path* γ is a continuous function over the closed interval $[0, 1]$. A *valid_path* is also piecewise-continuously differentiable. The operator *+++* joins two paths, yielding another path provided the endpoints meet.
- Functions *path_start* and *path_finish* return a path's start and finish, simply $\gamma(0)$ and $\gamma(1)$.
- The rectangle bounded by vectors *a* and *b* is written *cbox a b*, which for $\mathbb{R}^n$ is the set $\{x \mid a_i \leq x_i \leq b_i\}$, where v_i denotes the i-th component of a vector v. In the special case when the two vectors are real numbers *a* and *b*, the interval between them cab be written as *{a..b}*.

– Isabelle/HOL's analysis library has formalisations of the Lebesgue integral and the Henstock-Kurzweil gauge integral. [1] The predicate `set_integrable` takes a measure, a set, and a function and returns true if the function is Lebesgue integrable under the given measure on the given set. The function `borel_measurable` takes a measure and returns the set of functions that are measurable under that measure. The predicate `integrable_on` takes a set and a function and returns true if the Henstock-Kurzweil gauge integral of the function on the given set exists. The function `integral` takes a set and a function and returns the Henstock-Kurzweil gauge integral of the function on the given set, assuming that such integral exists. For more information on those constants please refer to [7, 8].

3 Basic Concepts and Lemmas

In this section we discuss the basic lemmas that we need to prove Green's theorem. However, we first need to discuss two basic definitions needed to state the theorem statement: *line integrals* and *partial derivatives*. Adapting these well-known concepts to Isabelle's analysis library required some thought and effort.

3.1 Line Integrals

For a vector field F defined on a Euclidean space, a parameterised path γ in the same Euclidean space, and a set of vectors B in the same Euclidean space, we define the line integral of F on γ as follows:

Definition 1. *Line Integral*

$$\int_{\gamma} F\downharpoonleft_B = \int_0^1 \sum_{b\in B} (F(\gamma(t))\cdot b)(\gamma'(t)\cdot b)dt$$

Above, b is a base vector from B, the symbol $\cdot$ denotes the inner product of two vectors, $F\downharpoonleft_B$ is the projection of F on the basis B, and the integral sign on the right hand side is the Henstock-Kurzweil gauge integral. A difference in our definition is that we add the argument B, a set of vectors, to which F and γ, and accordingly the line integral are projected. The reason for adding B is that we often refer to line integrals along a subset of base vectors, e.g. the integral of the x-component of a field along the x-component of a path. If we use the traditional formulation of line integrals (e.g. [17, p. 212]), we would need to pass the projections of both the field and the path, which is more cumbersome than passing the vectors on which we project once, as B. Formally, we need two definitions (one for the existence of the line integral):

```
definition line_integral::
 "('a::euclidean_space ⇒ 'a) ⇒ 'a set ⇒ (real ⇒ 'a) ⇒ real"
where
 "line_integral F basis γ ≡
    integral {0..1} (λx. ∑b∈basis. (F(γ (x)) · b)
           * (vector_derivative γ (at x within {0..1}) · b))"
```

[1] The gauge integral [15] is a generalisation of the well-known Riemann integral.

```
definition line_integral_exists
where
 "line_integral_exists F basis γ ≡
    (λx. ∑b∈basis. F(γ x) · b
          * (vector_derivative γ (at x within {0..1}) · b))
    integrable_on {0..1}"
```

In the above definition `(at x within s)` is an Isabelle/HOL mix infix notation for a filter (for more information on filters in Isabelle/HOL please consult [8]), where `x` is a point and `s` is a set. We use Isabelle's syntax capabilities to have $\int_\gamma F|_{basis}$ as well as $\oint_\gamma F|_{basis}$ denote `line_integral F basis γ`. For our definition of line integrals, the fundamental theorem of line integrals is as follows.

```
lemma fundamental_theorem_of_line_integrals_gen:
  fixes f::"'a::euclidean_space ⇒ real" and g::"real ⇒ 'a"
  assumes "∀a ∈ s. (has_gradient f f' a)"
    and "∀x ∈ {0..1}. γ x ∈ s"
    and "∀x ∈ {0..1}.
            (γ has_vector_derivative (γ' x)) (at x within {0..1})"
  shows "∫γ f' ↓Basis = f(γ 1) - f(γ 0)"
```

In the statement above the constant `has_gradient` indicates that `f` has the function `f'` as its gradient. Also, as one would expect, the line integral distributes over unions of disjoint sets of vectors and path joins as shown in the following statements.

```
lemma line_integral_sum_gen:
  assumes finite_basis:
    "finite basis" and
    line_integral_exists:
    "line_integral_exists F basis1 γ"
    "line_integral_exists F basis2 γ" and
    basis_partition:
    "basis1 ∪ basis2 = basis" "basis1 ∩ basis2 = {}"
  shows "∫γ F↓basis = ∫γ F↓basis1 + ∫γ F↓basis2"
    "line_integral_exists F basis γ"

lemma line_integral_distrib:
  assumes "line_integral_exists f basis γ1"
    "line_integral_exists f basis γ2"
    "valid_path γ1" "valid_path γ2"
  shows "∫γ1 +++ γ2 f↓basis =  ∫γ1 f↓basis + ∫γ2 f↓basis"
    "line_integral_exists f basis (γ1 +++ γ2)"
```

Line integrals also admit a transformation analogous to integration by substitution.

```
lemma line_integral_on_pair_path:
  fixes F::"'a::euclidean_space ⇒ 'a" and g::"real ⇒ 'a" and
      γ::"real ⇒ 'a" and i::'a
  assumes i_norm_1: "norm i = 1" and
   g_orthogonal_to_i: "∀x. g(x) · i = 0" and
   gamma_is_in_terms_of_i: "γ = (λx. f(x) *R i + g(f(x)))" and
   gamma_smooth: "γ C1_differentiable_on {0..1}" and
   g_continuous_on_f: "continuous_on (f ` {0..1}) g" and
   path_start_le_path_end: "(pathstart γ) · i ≤ (pathfinish γ) · i" and
   field_i_comp_cont: "continuous_on (path_image γ) (λx. F x · i)"
  shows "∫γ F ↓{i} =
                 integral {((pathstart γ) · i)..((pathfinish γ) · i)}
                          (λf_var. (F (f_var *R i + g(f_var)) · i))"
```

The lemma above applies to any path with all orthogonal components, but one (call it `i`), defined as a function `g` in terms of `i`.

3.2 Partial Derivatives

Partial derivatives are defined on the Euclidean space type class implemented in Isabelle/HOL. For a vector field F that maps a Euclidean space to another Euclidean space, we define its partial derivative to be w.r.t. the change in the magnitude of a component vector b of its input. At a point a, the partial derivative is defined as

Definition 2. *Partial Derivative*

$$\left.\frac{\partial F(v)}{\partial b}\right|_{v=a} = \left.\frac{dF(a + (x - a \cdot b)b)}{dx}\right|_{x=a\cdot b}$$

In the definition above, for a function f mapping reals to vectors, $\frac{df(x)}{dx}|_{x=c}$ denotes the vector derivative of f at the point where $x = c$. The Isabelle version of that definition above is as follows:

```
definition has_partial_vector_derivative::
 "('a::euclidean_space ⇒ 'b::euclidean_space) ⇒ 'a ⇒ 'b ⇒ 'a ⇒ bool"
where
 "has_partial_vector_derivative F b F' a
        ≡ ((λx. F(a - (a · b) *R b + x *R b ))
                has_vector_derivative F') (at (a · b))"

definition partially_vector_differentiable
where
 "partially_vector_differentiable F b p ≡
          (∃F'. has_partial_vector_derivative F b F' p)"

definition partial_vector_derivative::
 "('a::euclidean_space ⇒ 'b::euclidean_space) ⇒ 'a ⇒ 'a ⇒ 'b"
where
 "partial_vector_derivative F b a
        ≡ (vector_derivative (λx. F((a - ((a · b) *R b)) + x *R b))
          (at (a · b)))"
```

We also use Isabelle's syntax capabilities to have `∂ F / ∂i |a` denote `partial_vector_derivative F i a` and `∂ F / ∂i` denote `partial_vector_derivative F i`.

The definition that we use above resembles the *directional derivative*, which is a generalisation of the partial derivative. It is different from the partial derivative in that it is the change in the value of a function w.r.t. changes of its input in the direction of a given vector rather than the change in one of the variables on which F is defined. However, it is equivalent to the classical definition of a partial derivative when b is a base vector. This more general notion of derivative frequently allows us to remove the assumption that the given vector is a base vector. We also note that the following is an equivalent characterisation of that notion of derivatives, which can simplify some proofs.

```
lemma has_partial_vector_derivative_def_2:
"has_partial_vector_derivative F b F' a =
       ((λx. F(a + x *R b )) has_vector_derivative F') (at 0)"
```

The following result for the partial derivative follows from the fundamental theorem of calculus (FTC) for the vector derivative, proved in Isabelle/HOL analysis library.

```
lemma fundamental_theorem_of_calculus_partial_vector_gen:
  fixes k1 k2::"real" and
    F::"'a::euclidean_space ⇒ 'b::euclidean_space" and
    i::"'a" and
    F_i::"'a ⇒ 'b"
  assumes a_leq_b: "k1 ≤ k2" and
    unit_len: "i · i = 1" and
    no_i_component: "c · i = 0 " and
    has_partial_deriv:
      "∀p ∈ D. has_partial_vector_derivative F i (F_i p) p" and
    domain_subset_of_D:
      "{v. ∃x. k1 ≤ x ∧ x ≤ k2 ∧ v = x *R i + c} ⊆ D"
  shows "((λx. F_i( x *R i + c)) has_integral
                          F(k2 *R i + c) - F(k1 *R i + c)) {k1..k2}"
```

3.3 Green's Theorem for Elementary Regions

Given these definitions and basic lemmas, we can now start elaborating on our formalisation of Green's theorem. All proofs of Green's theorem that we encountered (e.g. Zorich and Cooke [17]) start by proving "half" of the theorem statement for every type of "elementary region" in $\mathbb{R}^2$. These regions are referred to as Type I or Type II regions, defined below.

Definition 3. *Elementary Regions*
A region D (modelled as a set of `real` *pairs) is Type I iff there are piecewise-C^1 smooth functions g_1 and g_2 such that for two constants a and b*

$$D_x = \{(x, y) \mid a \leq x \leq b \wedge g_2(x) \leq y \leq g_1(x)\}.$$

Similarly D would be called type II iff there are g_1, g_2, a and b

$$D_y = \{(x, y) \mid a \leq y \leq b \wedge g_2(y) \leq x \leq g_1(y)\}.$$

To prove Green's theorem, the typical approach is to prove the following two separate cases, for any regions D_x and D_y that are type I and type II, respectively, and their positively oriented boundaries:

$$\int_{D_x} -\frac{\partial(F_i)}{\partial j}\,dxdy = \int_{\partial D_x} F\downharpoonleft_{\{i\}},$$

and

$$\int_{D_y} \frac{\partial(F_j)}{\partial i}\,dxdy = \int_{\partial D_y} F\downharpoonleft_{\{j\}}.$$

Here i and j are the base vectors while F_i and F_j are the x-axis and y-axis components, respectively, of F. The difference in the expressions for the type I and type II regions is because of the skew symmetry of the x-axis and the y-axis w.r.t. the orientation. We refer to the top expression as the *x-axis* Green theorem, and the bottom one as the *y-axis* Green theorem.

To avoid having near-duplicate proofs, one for the x-axis and another for the y-axis, we formulate a locale *i_j_orthonorm* that treats arbitrary orthonormal unit vectors *i* and *j*, for which there is a Fubini like theorem (i.e. *i* and *j* commute under iterated integration). A *locale* is a named context: definitions and theorems proved within locale *i_j_orthonorm* can refer to the variables and assumptions declared there. Within that locale, we prove the following Isabelle/HOL statement, which can be seen as the x-axis Green theorem for type I regions (if *i* is assigned to be the x-axis and *j* is assigned to be the y-axis). For the boundary, we model its paths explicitly as functions of type *real ⇒ 'a::euclidean_space*, where *γ1*, *γ2*, *γ3* and *γ4* are the bottom, right, top and left sides, respectively. Also, from now on we let *(F i)* denote *(λa. F a . i)*, i.e. the *i*th component of the field *F*.

```
lemma GreenThm_type_I:
  fixes F and
    γ1 γ2 γ3 γ4 and
    a::"real" and b::"real" and
    g1::"real ⇒ real" and g2::"real ⇒ real"
  assumes
    "Dy_pair = {p. ∃x y. p = x *R i + y *R j ∧
                         x ∈ {a..b} ∪ {b..a} ∧
                         y ∈ {(g2 x)..(g1 x)} ∪ {(g1 x)..(g2 x)}}"
    "γ1 = (λx. let c = (linepath a b x) in c *R i + g2 c *R j)"
    "γ1 piecewise_C1_differentiable_on {0..1}"
    "γ2 = (λx. b *R i + (linepath (g2 b) (g1 b) x) *R j)"
    "γ3 = (λx. let c = (linepath a b x) in c *R i + g1 c *R j)"
    "γ3 piecewise_C1_differentiable_on {0..1}"
    "γ4 = (λx. a *R i + (linepath (g2 a) (g1 a) x) *R j)"
    "analytically_valid Dy_pair (F i) j i"
    "(∀x ∈ {a..b} ∪ {b..a}.
        (g2 x) ≤ (g1 x)) ∨ (∀x ∈ {a..b} ∪ {b..a}. (g1 x) ≤ (g2 x))"
    "a ≠ b"
  shows "∫γ1 F ↓{i} + ∫γ2 F ↓{i} - ∫γ3 F ↓{i} - ∫γ4 F ↓{i}
           = (if a < b then 1::int else -1) *
               (if (∀x ∈ {a..b} ∪ {b..a}. (g2 x) ≤ (g1 x))
                  then 1::int
                  else -1) *
                 integral Dy_pair (λa. - (∂ (F i)/ ∂j |a))"
```

Proving the lemma above depends on the observation that for a path γ (e.g. *γ1* above) that is orthogonal to a vector i, $\int_\gamma F\downharpoonright_{\{i\}} = 0$, for an F continuous on γ.[2] The rest of the proof boils down to an application of Fubini's theorem (which is assumed to hold for *i* and *j*) and the FTC to the double integral, the integral by substitution to the line integrals and some algebraic manipulation [17, p. 238]. Nonetheless, this algebraic manipulation proved to be quite tedious when done formally in Isabelle/HOL.

[2] Formally, this observation follows immediately from theorem *line_integral_on_pair_path*.

3.4 Our Analytic Assumptions

The predicate `analytically_valid` in the last lemma represents the analytic assumptions of our statement of Green's theorem, to which an "appropriate" field has to conform. Firstly let 1_s be the indicator function for a set s. Then, for a type I region D_x our analytic assumptions for the x-axis Green theorem are that

(i) F_i is continuous on D_x
(ii) $\frac{\partial(F_i)}{\partial j}$ exists everywhere in D_x
(iii) the product $1_{D_x}(x, y)\frac{\partial(F_i)}{\partial j}(x, y)$ is Lebesgue integrable
(iv) the product $1_{[a,b]}(x)\int_{g_1(x)}^{g_2(x)} F(x, y)dy$ is a Borel measurable function, where the integral in that function is a Henstock-Kurzweil gauge integral.

These assumptions vary symmetrically for the y-axis Green theorem, so to avoid having two symmetrical definitions, we define the predicate `analytically_valid` to take the axes as arguments.

```
definition analytically_valid
where
 "analytically_valid s F i j ≡
       continuous_on s F ∧
       (∀a ∈ s. partially_vector_differentiable F i a) ∧
       set_integrable lborel s (∂ F / ∂ i) ∧
       (let p = (λx y. (y *R i + x *R j)) in
            ((λx. integral UNIV
                     (λy. (indicator s (p x y)) *R (∂ F / ∂ i |p x y)))
             ∈ borel_measurable lborel))"
```

These conditions refer to Lebesgue integrability and to measurability because we use Fubini's theorem for the Lebesgue integral in Isabelle/HOL's Analysis library to derive a Fubini like result for the Henstock-Kurzweil integral. Note that proving Fubini's theorem for the gauge integral would allow for more general analytic assumptions, and we hope to do this eventually.

4 The Treatment of More General Regions

Now that we have described some of the basic definitions and how to derive Green's theorem for elementary regions, the remaining question is how to prove the theorem for more general regions. As we stated in the introduction, textbook proofs of Green's theorem typically require regions that can be divided into elementary regions. It can be shown that any *regular* region can be divided into elementary regions [12, 17]. Regular regions (as defined in Protter [12, p. 235]) are regions whose boundaries are piecewise-smooth. Practically, those regions are enough for most applications, especially in physics and engineering [17].

In this section we describe how we prove Green's theorem for regions that can be divided into both type I regions and type II regions using *only* vertical and horizontal edges, respectively. We believe that for most practical purposes, the additional



Fig. 1: An annulus and its division in type I and type II regions. Every 1-cube (i.e. path) is represented by an arrow whose direction is the same as the orientation of the 1-cube. a) The positively oriented boundary of the annulus. b), c) and d) The members of a type I division of the annulus. e) A 1-chain that includes all the horizontal boundaries in the type I division. f), g) and h) The members of a type II division of the annulus. i) A 1-chain that includes all the vertical boundaries in the type II division. j) A common subdivision of the chains in e) and i).

assumption that the division is done only by vertical and horizontal edges is equivalent to assuming just the existence of type I and type II divisions. Figure 1 shows an example of a region and its type I and type II divisions. In this example, some of the elementary regions appear to have a missing edge. This is because the type I or the type II divisions induced a *one-point path*: a function mapping the interval $[0, 1]$ to a single point in $\mathbb{R}^2$. For instance, the left edge in the left 1-chain in 1b is a point on the x-axis.

4.1 Chains and cubes

For tackling more general regions and their boundaries we use the concepts of cubes and chains [14, chapter 8]. One use of cubes is to represent parameterisable surfaces (regions in $\mathbb{R}^2$ and paths in our case). A k-dimensional such surface embedded in $\mathbb{R}^n$ is represented by a function whose domain is a space homeomorphic to $\mathbb{R}^k$ and whose codomain is $\mathbb{R}^n$. Roughly speaking, we model cubes as functions and chains as sets of cubes. We use the existing Isabelle/HOL formalisation of paths, where we model 1-cubes as functions defined on the interval `{0..1}`. We model a 1-chain as a set of pairs of `int` (coefficients) and 1-cubes.

```
type_synonym path = "real ⇒ (real * real)"

type_synonym one_chain = "(int * path) set"
```

The following definition is an example that shows how the above concepts are used to lift line integrals to 1-chains.

```
definition one_chain_line_integral::
 "(real * real => real * real) => (real * real set) ⇒ one_chain ⇒ real"
where
 "one_chain_line_integral F b C =
              (∑ (k,γ)∈C. k * (line_integral F b γ))"
```

We also use Isabelle's syntax capabilities to have $^{H}\int_{\gamma} F\downharpoonright_{basis}$ as well as $^{H}\oint_{\gamma} F\downharpoonright_{basis}$ denote `one_chain_line_integral F basis γ`.

Note that our notion of chains is different from chains as they are usually defined in the mathematics literature. In traditional expositions, chains are built on the concept of formal sums. This allows for mutual cancellation of cubes of the same form but opposite orientations. Although this could easily be implemented using multisets, we omit it as it is practically irrelevant according to our experience.

We extend the way we model 1-cubes to model 2-cubes, which we model as functions of type `(real * real ⇒ real * real)`. These functions are defined on the rectangle `cbox (0,0) (1,1)`, to which we refer as `unit_cube`.

```
type_synonym two_cube = "(real * real ⇒ real * real)"

definition cubeImage
where
 "cubeImage twoC ≡ (twoC ` unit_cube)"
```

The orientation of the boundary of a 2-cube (a 1-chain) is taken to be anticlockwise. A 1-cube is given the coefficient -1 if the path's direction is clockwise, else 1. Formally this is defined as follows:

```
fun horizontal_boundary::"two_cube ⇒ one_chain"
where
 "horizontal_boundary C = {(1, (λx. C(x,0))), (-1, (λx. C(x,1)))}"

fun vertical_boundary::"two_cube ⇒ one_chain"
where
 "vertical_boundary C = {(-1, (λy. C(0,y))), (1, (λy. C(1,y)))}"

definition boundary::"two_cube ⇒ one_chain"
where
 "boundary C = horizontal_boundary C ∪ vertical_boundary C"
```

For boundaries we use the Isabelle syntax capabilities to have ∂ denote `boundary`. We follow the convention in defining the 2-cubes in such a way that the top and left edges are against the anticlockwise orientation (e.g. see the 2-cube in Figure 1c). Accordingly both the left and top edges take a -1 coefficient in the 1-cube representation. This leads to simpler formal definitions of type I and type II 2-cubes. However, to avoid repeating symmetric definitions, we define a type I region with respect to two argument vectors `v1` and `v2`. We also pass an additional orientation argument which is an integer whose sign indicates whether the given 2-cube conforms to the orientation convention (i.e. if its top and left edges are against the anticlockwise orientation, and bottom and right edges are aligned with the anticlockwise orientation). That orientation argument is necessary to enable the usage of this predicate with both type I and type II 2-cubes.

```
fun typeI_twoCube
where
 "typeI_twoCube (orient::int, C) v1 v2 =
  (∃a b g1 g2. a ≠ b ∧
    ((∀x ∈ ({a..b} ∪ {b..a}). g2 x ≤ g1 x) ∨
      (∀x ∈ ({a..b} ∪ {b..a}). g1 x ≤ g2 x)) ∧
    C = (λ(t1,t2). let c = (linepath a b t1) in
                       (c *R v1 + linepath (g2 c) (g1 c) t2 *R v2)) ∧
    g1 piecewise_C1_differentiable_on ({a..b} ∪ {b..a}) ∧
    g2 piecewise_C1_differentiable_on ({a..b} ∪ {b..a}) ∧
    orient = (if a < b then 1 else -1) *
              (if ∀x∈{a..b} ∪ {b..a}. g2 x ≤ g1 x then 1 else - 1))"
```

For instance, to indicate a 2-cube is a type I region using the predicate above, `v1` has to be assigned with the x-axis and `v2` has to be assigned with the y-axis, and the orientation argument has to assigned based on the orientation of the 2-cube.

We also require that all 2-cubes conform to the following predicate:

```
definition valid_two_cube
where
 "valid_two_cube twoC = (card (boundary twoC) = 4)"
```

This predicate filters out cases where 2-cubes have either: i) right and top edges that are both one point paths, or ii) left and bottom edges that are both one point paths. Although this assumption potentially leaves our theorems less general regarding some corner cases, it makes our computations much smoother. After defining these concepts on 2-cubes, we derive the following statement of Green's theorem in terms of 2-cubes.

```
lemma GreenThm_typeI_twoCube:
  shows "H∮∂C F↓{i} =
                orient * integral (cubeImage C) (λp. - ∂ Fi / ∂ j |p)"
    "∀(k,γ) ∈ ∂C. line_integral_exists F {i} γ"
```

We note that this statement is derived within the following locale, `green_typeI_cube`, that identifies conditions on the 2-cube. Accordingly, it can be instantiated to either the x-axis Green theorem or the y-axis Green theorem, by appropriately assigning the vectors `i` and `j` which are implicitly fixed in `green_typeI_cube` since that locale includes `i_j_orthonorm`.

```
locale green_typeI_cube = i_j_orthonorm +
  fixes C and orient and F::"real * real ⇒ real * real"
  assumes
    two_cube: "typeI_twoCube (orient, C) i j" and
    valid_two_cube: "valid_two_cube (C)" and
    f_analytically_valid: "analytically_valid (cubeImage C) (Fi) j i"
```

Although we anticipated that proving `GreenThm_typeI_twoCube` would be a straightforward unfolding of definitions and usage of `GreenThm_typeI`, it was a surprisingly long and tedious proof that took a few hundred lines.

For 2-chains, we model them as sets of pairs of integers and 2-cubes as follows.

```
type_synonym two_chain = "(int * two_cube) set"
```

We define the boundary of a 2-chain as follows:

```
definition two_chain_boundary:: "two_chain ⇒ one_chain"
where
 "two_chain_boundary twoChain = ⋃((boundary o snd) ` twoChain)"
```

We similarly defined the functions `two_chain_horizontal_boundary` and `two_chain_vertical_boundary`. We also lift the double integral to 2-chains as follows.

```
definition two_chain_integral::
"two_chain ⇒ (real * real ⇒ real) ⇒ real"
where
 "two_chain_integral twoChain F ≡
        ∑(orient, C)∈twoChain. orient * (integral (cubeImage C) F)"
```

Lastly, to smooth our computations on integrals over 2-chains and their boundaries, we require that a 2-chain

(i) has all members be valid 2-cubes,
(ii) edges of different 2-cubes have equal images only if they have opposite orientations,
(iii) different 2-cubes have different images, and
(iv) all cubes within the chain have consistent orientations.

Again we believe that these requirements will only rule out corner cases that are practically irrelevant. The requirements are formally defined in the following predicate:

```
definition valid_two_chain
where
 "valid_two_chain twoChain ≡
    (∀(orient, C) ∈ twoChain. valid_two_cube C)
    ∧ pairwise (λc1 c2. (∂ (snd c1)) ∩ (∂ (snd c2)) = {}) twoChain
    ∧ inj_on (cubeImage o snd) twoChain
    ∧ (∃orient. ∀(orient',C) ∈ twoChain. orient' = orient)"
```

Note: `two_chain_boundary` is only intended to be used with 2-chains that satisfy `valid_two_chain`. For instance, a 2-chain that violates `two_chain_boundary` can have a boundary with multiple occurrences of the same 1-cube with the same orientation. In this case only one of those cubes will be present in the 1-chain returned by `two_chain_boundary`, and thus the multiplicity of those repeated cubes will be lost.

Given these definitions on 2-chains, we lift our statement of Green's theorem from 2-cubes to 2-chains, as shown in the following statement.

```
lemma GreenThm_typeI_twoChain:
  shows "H∮(two_chain_boundary two_chain) F ↓{i} =
            two_chain_integral two_chain (λp. - ∂ F_i / ∂j |p)"
```

Again, this statement is proved within the following locale that identifies the appropriate conditions on the 2-chain.

```
locale green_typeI_chain = i_j_orthonorm +
  fixes F::"real * real ⇒ real * real" and two_chain s
  assumes "valid_typeI_division s two_chain i j" and
         "∀(orient,C) ∈ two_chain.
                    analytically_valid (cubeImage C) (F_i) j i"
```

In the previous locale, `valid_typeI_division` is an abbreviation that, for a region and a 2-chain, means that the 2-chain constitutes only valid type I cubes (w.r.t. the given vectors) and that this 2-chain is a division of the given region.

After proving `GreenThm_typeI_twoChain`, the next step is to instantiate it with the proper unit vectors representing type I divisions and type II divisions to obtain the x-axis and the y-axis Green statements for 2-chains.

```
locale green_typeI_typeII_chain =
      i_j_orthonorm: i_j_orthonorm i j +
      T1: green_typeI_chain i j F two_chain_typeI +
      T2: green_typeI_chain "-j" i F two_chain_typeII
    for i j neg_j F two_chain_typeI two_chain_typeII
```

In the locale above, the instantiation `T1` of `green_typeI_chain` yields the x-axis Green theorem for 2-chains and the instantiation `T2` yields the y-axis statement.

After obtaining the x-axis and y-axis statements for 2-chains and after some algebraic and analytic manipulation, the next step is to add the line integral sides of the x-axis Green theorem to its counterpart in the y-axis theorem and similarly add the double integrals of both theorems. Directly adding up both sides of the equalities in the conclusion can give us Green's theorem in terms of 2-chains and their boundaries. However, the main goal of the paper is to obtain the theorem directly for a region and its boundary, assuming that the region can be *vertically* sliced into subregions of type I and *horizontally* sliced into subregions of type II.

The first (and easier) part in proving this is to prove the equivalence of the double integral on a region and the integral on a 2-chain that divides that region. Before deriving such a theorem we generalised the notion of `division_of`, defined in Isabelle/HOL's multivariate analysis library, to work when the division is not constituted of rectangles.

```
definition gen_division
where
 "gen_division s S ≡
    (finite S ∧ (⋃S = s) ∧ (pairwise (λu t. negligible (u ∩ t))) S)"
```

Then we show the following equivalence:

```
lemma two_chain_integral_eq_integral_divisable:
  assumes "∀(orient, C) ∈ twoChain. F integrable_on cubeImage C" and
    "gen_division s ((cubeImage o snd) ` twoChain)" and
    "valid_two_chain twoChain"
  shows "two_chain_integral twoChain F =
            (chain_orientation twoChain) * (integral s F)"
```

In the above theorem the 2-chain orientation is defined as follows.

```
definition "chain_orientation twoChain =
            (THE orient. ∀(orient',C) ∈ twoChain. orient' = orient)"
```

The other part, concerning the line integrals, proved to be trickier. We explain this in the next section.

4.2 Dealing with Boundaries

What remains now is to prove an equivalence between the line integral on the 1-chain boundary of the region under consideration and the line integral on the 1-chain boundary of the region's elementary divisions (i.e. the 2-chain division of the region). The classical approach reasons that the line integrals on the introduced boundaries will cancel each other out, leaving the line integral on the region's original boundary. For example, the vertical-straight-line paths in Figures 1b, 1c and 1d, are the introduced boundaries to obtain the type I division of the annulus. In this example, the line integrals on the introduced vertical-straight-line paths cancel because of their opposite orientations.

To prove this formally, the classical approach [3,12] needs to define a *positively oriented boundary*, which requires an explicit definition of the boundary of a region, and also defining the exterior normal of the region. However, we use a different approach that does not depend on these definitions and avoids a lot of the resulting geometrical and analytic complications. Our approach depends on two observations:

O1 If a path γ is orthogonal to a vector i, then $\int_\gamma F\downharpoonright_{\{i\}} = 0$, for F continuous on γ.

O2 Dividing the region in type I/type II regions is done by introducing only vertical/horizontal boundaries.

For a type I 2-chain division of a region, consider the 1-chain γ_x, that: i) includes *all* the horizontal boundaries of the dividing 2-chain, and ii) includes *some* subpaths of the vertical boundaries of the dividing 2-chain (call this condition C). Based on O1, the line integral on the vertical edges in the 1-chain boundary of the type I division and accordingly γ_x, projected on i, will be zero. Accordingly we can prove the x-axis Green theorem for γ_x. An analogous condition for a type II division asserts that the 1-chain includes *all* the vertical boundaries of the dividing 2-chain, and includes *some* subpaths of the horizontal boundaries of the dividing 2-chain. Continuing with our approach to exploit symmetries, we formalise the argument for obtaining the x-axis statement w.r.t. two orthonormal vectors, and then instantiate those vectors appropriately to get either the x-axis or the y-axis statement.

Formal statements of C, and the consequence of a 1-chain conforming to it, come next.

```
lemma GreenThm_typeI_divisible_region_boundary_gen:
  assumes only_vertical_division: "only_vertical_division γ two_chain"
  shows "H∫γ F ↓{i} =
    (chain_orientation two_chain) * integral s (λp. - (∂ Fi / ∂ j |p))"
```

In the above statement, the condition C is stated as the predicate `only_vertical_division` relating a 1-chain to a region, and it is defined as follows.

```
definition only_vertical_division
where
 "only_vertical_division one_chain two_chain =
  (∃𝒱 ℋ. finite ℋ ∧ finite 𝒱 ∧ boundary_chain ℋ ∧ one_chain = 𝒱 ∪ ℋ
  ∧ (∀(k, γ) ∈ 𝒱.
     (∃ (k', γ') ∈ two_chain_vertical_boundary two_chain.
     (∃ a ∈ {0..1}. ∃ b ∈ {0..1}. a ≤ b ∧ subpath a b γ' = γ))) ∧
 (common_subdiv_exists (two_chain_horizontal_boundary two_chain) ℋ
 ∨ common_reparam_exists ℋ (two_chain_horizontal_boundary two_chain))"
```

Note that in `only_vertical_division` the two predicates `common_subdiv_exists` and `common_reparam_exists` are two ways to characterise equivalence of 1-chains via the existence of common subdivisions between them. We discuss the details of the formalisation of those two predicates below.

The lemma above is in the locale `green_typeI_chain`, so there will be two different instances of it in the locale `green_typeI_typeII_chain`, one in `T1` and another in `T2`, that can be interpreted as the x-axis and the y-axis versions of the lemma, respectively.[3] Note also that in the `T2`-instance of this lemma, the division will be type II and the hypothesis `only_vertical_division` will mean that this division is obtained by only adding **horizontal** lines, in contrast to the `T1`-instance. This is because the axes are rotated in `T2`.

From the second observation, O2, we can conclude that there will always be

- a 1-chain, γ_x, whose image is the boundary of the region under consideration and that satisfies C for the type I division.
- a 1-chain, γ_y, whose image is the boundary of the region under consideration and that satisfies C for the type II division, where it is *not* necessary that $\gamma_x = \gamma_y$.

Figure 1e and Figure 1i show two 1-chains that satisfy C for the type I and type II divisions of the annulus. Notice that in this example, those two 1-chains are *not equal* even though they have the same *orientation* and *image*.

Now, if we can state and formalise the equivalence between γ_x and γ_y, and that this equivalence lifts to equal line integrals, we can obtain Green's theorem in terms of the region, which is our goal. One way to formalise path equivalence is to explicitly define the notion of orientation. Then the equivalence between γ_x and γ_y can be characterised by their having similar orientations and images. An advantage of this approach is that it can capture equivalence in path orientations regardless of the path image.

However, we do not need this generality in the context of proving the equivalence of 1-chains that have the same image and orientation, especially that this generality will cost a lot of analytic and geometric complexities to be formalised. Instead we choose to formalise the notion of equivalence in terms of having a *common subdivision*, where we define two notions of subdivisions, which both imply equivalence of line integrals. For example the 1-chain shown in Figure 1j is a subdivision of each of the 1-chains in Figure 1e and Figure 1i as well as the original boundary 1-chain in Figure 1a, i.e. a common subdivision between the three 1-chains.

The first type of a common subdivision is formalised by effectively stating that the two 1-chains can each be "recursively" joined to form two paths with the *equal* parameterisations. The second type of common subdivisions that we define informally states that one of the two 1-chains is made of paths that are "reparameterisations" of the paths in the other 1-chain. The first type of 1-chain equivalence is easier to deal with in practical examples (including the diamond example in the last section). However, the reparameterisation based path equivalence is more general and thus applies to more parameterisations. We now describe in details both types of common subdivisions.

[3] This is not exactly true, since the instantiations `T1` and `T2` are obtained using a pair of orthonormal unit vectors `i` and `j`. If `i` and `j` are to be assigned to `(1,0)` and `(0,1)`, then the instantiations of `GreenThm_typeI_divisible_region_boundary_gen` in `T1` and `T2` can be seen as the x-axis and y-axis Green statements.

For both concepts of common subdivisions between 1-chains, we focus on “boundary” 1-chains, defined as follows.

```
definition boundary_chain
where
 "boundary_chain s = (∀ (k, γ) ∈ s. k = 1 ∨ k = -1)"
```

We now discuss the first type of common subdivisions, defined as follows.

```
inductive chain_subdiv_path
where
 singleton: "chain_subdiv_path (coeff_cube_to_path γ) {γ}" |
 insert:
    "γ ∉ s ⟹ pathfinish (coeff_cube_to_path γ) = pathstart γ0
      ⟹ chain_subdiv_path γ0 s ⟹
       chain_subdiv_path (coeff_cube_to_path γ +++ γ0) (insert γ s)"
```

It lifts the `path_join` operator defined in the Isabelle/HOL multivariate analysis library, to act on 1-chains ordered into lists. A necessary condition for the joined paths to be usable for our purposes is that the ending point of every path is the starting point of the next. Note that in the above definition `coeff_cube_to_path` is a function that takes an integer and a path and returns the path if the integer is positive, and its inverse otherwise.

We call a 1-chain γ, a subdivision of another 1-chain η, if one can map every cube in η to a sub-chain of γ that is a subdivision of it. Formally this is defined as follows:

```
definition chain_subdiv_chain
where
 "chain_subdiv_chain one_chain1 subdiv ≡
  ∃ f. (⋃(f ` one_chain1)) = subdiv ∧
     (∀ c∈one_chain1. chain_subdiv_path (coeff_cube_to_path c) (f c)) ∧
      pairwise (λ p p'. f p ∩ f p' = {}) one_chain1"
```

After proving that each of the previous notions of equivalence implies equality of line integrals, we define equivalence of 1-chains in terms of having a common subdivision, and prove that it implies equal line integrals. We define it as having a boundary 1-chain that is a subdivision for each of the 1-chains under consideration, which is formally stated as follows.

```
definition common_subdiv_exists
where
 "common_subdiv_exists one_chain1 one_chain2 =
   (∃ subdiv ps1 ps2. chain_subdiv_chain (one_chain1 - ps1) subdiv ∧
       chain_subdiv_chain (one_chain2 - ps2) subdiv ∧
       (∀ (k, γ) ∈ subdiv. valid_path γ) ∧ (boundary_chain subdiv) ∧
       (∀ (k, γ) ∈ ps1. point_path γ) ∧ (∀ (k, γ) ∈ ps2. point_path γ))"
```

The following statement shows the equality of line integrals implied by `common_subdiv_exists`, and the other conditions needed for it.

```
lemma gen_common_subdivision_imp_eq_line_integral:
  assumes "(common_subdiv_exists one_chain1 one_chain2)"
    "boundary_chain one_chain1"
    "boundary_chain one_chain2"
    "∀ (k, γ)∈one_chain1. line_integral_exists F basis γ"
    "finite one_chain1" "finite one_chain2" "finite basis"
  shows "^H∫ one_chain1 F ↾basis = ^H∫ one_chain2 F ↾basis "
    "(k, γ)∈one_chain2 ⟹ line_integral_exists F basis γ"
```

Defining the second type of common subdivisions depends on the following relation between paths, that states the conditions under which a path can be considered a reparameterisation of another path. The main condition is that one path is the composition of the other with a piecewise-smooth map that is bijective on `{0..1}`.

```
definition reparam
where
 "reparam γ1 γ2 ≡
   ∃φ. (∀x∈{0..1}. γ1 x = (γ2 o φ) x) ∧ bij_betw φ {0..1} {0..1} ∧
      φ piecewise_C1_differentiable_on {0..1} ∧ φ(0) = 0 ∧ φ(1) = 1 ∧
      φ -` {0..1} ⊆ {0..1}"
```

Based on that we define the following equivalence relation between a chain and a path, that is analogous to `chain_subdiv_path`.

```
definition chain_reparam_path
where
 "chain_reparam_path γ one_chain ≡
     ∃γ'. chain_subdiv_path γ' one_chain ∧ reparam γ γ'"
```

That relation is also lifted to chains in a similar manner to the subdivision relation. We also use the lifted relation to define a reparameterisation based concept of common subdivision which implies the equivalence of line integrals.

4.3 The Formalised Statement of Green's Theorem

Based on the previous concept of common subdivision and common reparameterisation we finally prove the following statement of Green's theorem. This theorem was derived by combining the two instantiations of `GreenThm_typeI_ divisible_region_ boundary_gen` that come from `T1` and `T2` in the locale `green_ typeI_typeII_chain`, i.e. it comes from the lemmas for type I and type II regions. Note that if in the instantiation `T2` the vector `-j` is replaced by `j`, we would instead obtain the divergence theorem.

```
lemma GreenThm_typeI_typeII_divisible_region_anti_cwise:
  assumes
    only_vertical_division:
    "only_vertical_division one_chain_typeI two_chain_typeI"
    "boundary_chain one_chain_typeI" and
    only_horizontal_division:
    "only_vertical_division one_chain_typeII two_chain_typeII"
    "boundary_chain one_chain_typeII" and
    typeI_and_typII_one_chains_have_gen_common_subdiv:
    "common_subdiv_exists one_chain_typeI one_chain_typeII" and
    same_orientation:
    "chain_orientation two_chain_typeI =
           chain_orientation two_chain_typeII"
    "chain_orientation two_chain_typeII = 1"
  shows "H∫ one_chain_typeI F ↓{i, j} =
                              integral s (λx. ∂ F_j/ ∂i |x - ∂ F_i/ ∂j |x)"
        "H∫ one_chain_typeII F ↓{i, j} =
                              integral s (λx. ∂ F_j/ ∂i |x - ∂ F_i/ ∂j |x)"
```

The theorem above does not require the 1-chains γ_x and γ_y to have as their image exactly the boundary of the region. However, of course it applies to the 1-chains if their image is the boundary of the region. Accordingly it fits as Green's theorem for a region that can be divided into elementary regions just by vertical and horizontal slicing.

It is worth noting that although this statement seems to have a lot of assumptions, its analytic assumptions regarding the field are strictly more general than those in standard textbooks [12, 17], where they require the field and both of its partial derivatives to be continuous in the region. The following statement shows that our analytic assumptions are at least as general.

```
lemma C1_imp_analytically_valid_ij:
 assumes "continuous_on (s×t) F" "compact t" "s = {a..b}"
         "C1_partially_differentiable_on F (s×t) i"
  shows "analytically_valid (s×t) F i j"
```

In practice, we believe that it might be easier to show continuous partial differentiability as defined below, than directly showing that a field is analytically valid.

```
definition C1_partially_differentiable_on
where
 "C1_partially_differentiable_on C s b =
        (∃C'. (∀x∈ s. (has_partial_vector_derivative C b (C' x) x)) ∧
               continuous_on s C')"
```

For the geometric assumptions, on the other hand, we have two extra requirements compared to typical textbook statements of the theorem: the type I and type II divisions should be obtained using only vertical slicing and only horizontal slicing, respectively. We conjecture that those extra assumptions are innocuous:

Conjecture 1. *If a region in the plane can be divided into finitely many type I regions, then it can be divided into finitely many type I regions by introducing only vertical boundaries. Similarly, if a region in the plane can be divided into finitely many type II regions, then it can be divided into finitely many type II regions by introducing only horizontal boundaries.*

We note that removing the extra assumption (namely, that a common subdivision or reparameterisation exists) from the theorem is not only a matter of resolving the geometric question in Conjecture 1. Those assumptions are inherent to our approach of stating the theorem, because we represent the region's boundary using a 1-chain whose relation to the region is asserted only by assuming that this 1-chain is a reparameterisation of the horizontal (vertical) boundaries of the type I (type II) divisions. If the conjecture is resolved, then the assumption that the type I (type II) division was obtained only through inserting vertical (horizontal) boundaries could be replaced by an assumption that the boundaries of the type I and type II divisions only share a common subdivision or reparameterisation. However, in this case the type I (type II) boundaries could include some horizontal (vertical) introduced boundaries, and one would have to cancel out line integrals on the introduced horizontal (vertical) boundaries in every application of the theorem. This will make applying the theorem to examples cumbersome, since cancelling out intermediate boundaries in addition to constructing the type I and type II divisions would be substantially more complicated than only constructing type I and type II divisions by adding only vertical and horizontal lines, respectively. The other possibility is that we remove the extra geometric assumptions from the theorem statement and use an explicit specification of the boundary with the aid of an outwards pointing normal vector. Nonetheless, using this explicit representation of the boundary in the formalisation will make the proofs substantially more complicated, which is precisely why we adopted the current representation of the boundary.

In the next section we apply the theorem to two regions: a diamond and a disk. We show that proving the extra geometric assumptions for applications is a reasonable task.

5 Usability of the Formalised Statement

In this section we investigate the applicability of the statement that we formalised, and the *practical* impact of the extra geometric assumptions. We first derive the following special case of the theorem.

```
lemma GreenThm_typeI_typeII_divisible_region_finite_holes:
 assumes valid_cube_boundary: "∀(k,γ)∈boundary C. valid_path γ" and
    only_vertical_division:
    "only_vertical_division (∂C) two_chain_typeI" and
    only_horizontal_division:
    "only_vertical_division (∂C) two_chain_typeII" and
    s_is_oneCube: "s = cubeImage (C)" and
    "two_chain_typeI = (orient, C)" "two_chain_typeII = (orient, C)"
shows
"H∮∂C F ↓{i, j} =

   orient *
      integral (cubeImage C) (λx. (∂ F_j / ∂ i |x) - (∂ F_i / ∂ j |x))"
```

We derived it based on the reflexivity properties of the different subdivision and reparameterisation constructs that we defined earlier. Except for the extra geometric assumptions, the statement above resembles the statement of Green's theorem as it would be rendered in classical textbook treatments. It directly relates the double integral on a region to the line integral on the region's boundary, without using an intermediate subdivision like we did in the more general statement *GreenThm_typeI_typeII_divisible_region*, and with minimal reference to the homological concepts of chains and cubes. This special case can be used primarily for regions without holes, like the regions considered by Harrison in his formalisation of Cauchy's integral theorem [5]. According to Protter [12], the first and second geometric assumptions are equivalent to assuming that the region has piecewise-smooth boundaries.

5.1 Two Small Examples

To demonstrate the practical utility of our formalised statement we apply it to two regions: a diamond and a disk. In particular, we aim to study the complexity of proving that the geometric assumptions apply to realistic examples. The reason motivating



Fig. 2: (a), (b), (c), and (d) show the left, top, bottom boundaries of the type I diamond. The left and right edges are degenerate straight lines i.e. points, thus qualifying it to be a type I 2-cube. (e), (f), (g), and (h) show the left, top, bottom boundaries of the type II diamond. The top and bottom edges are degenerate straight lines i.e. points, thus qualifying it to be a type II 2-cube. (i), (j), (k), and (l) show four 1-cubes that are a common subdivision between the non-degenerate part of the type I diamond's boundary (i.e. (c) and (d)) and the vertical part of the type II diamond's boundary. Note that their orientations are different, but this is accounted for by the integer coefficients in the chain representation.

our focus on the geometric assumptions is that this is where the statement that we formalised is different with respect to other mainstream statements of the theorem. The first step is to formulate the diamond and the disk as 2-cubes. We do so as follows, where the real number `d` is the diameter for both shapes.

```
definition diamond_cube
where
 "diamond_cube =
    (λ(t1,t2). let x = (t1 - 1/2) * d in
                   (x, (2 * t2 - 1) * (d/2 - |x|)))"
```

```
definition disk_cube
where
 "disk_cube =
    (λ(t1,t2). let x = (t1 - 1/2) in
                (x * d, (2 * t2 - 1) * d * sqrt (1/4 - x * x)))"
```

It should be clear that both of those parameterisations are type I, because the y-coordinates for both are functions in the x-coordinate. For the diamond, the four boundaries are shown in Figure 2. The top and bottom edges are each comprised of two line segments and the right and left edges are point paths (i.e. degenerate lines) as shown in Figures 2a-2d. For the disk, the top and bottom edges are arcs instead of line segments, while the right and left edges are points too.

The first geometric assumption we need the diamond and the disk to conform to is that each of their four boundaries is piecewise differentiable. Proving this was straightforward for the diamond since the four boundaries are comprised of straight lines and points, while for the disk it was slightly more tedious because of the presence of the square roots.

The second geometric assumption that we need to show for the regions is that they admit a type I division obtained by only vertical slicing. The type I division that we use is a singleton: it is the parameterisation itself, i.e. `{diamond_cube d}` for the diamond and `{disk_cube d}` for the disk, which are both type I. Proving that these type I parameterisations are obtained only via introducing only vertical boundaries is easy, since we add no boundaries.

Next, we need to show the existence of a type II subdivision that can be obtained by introducing *only* horizontal boundaries. Since both the diamond and disk are symmetric, the type II parameterisations can be the type I parameterisations but after i) reversing their orientation and ii) reflecting them on the identity line. The way to obtain type II parameterisations is shown below.

```
definition typeI_to_typeII
where
 "typeI_to_typeII C = prod.swap o C o (λ(x,y). (1 - x, y))"
```

In the above definition, the function `prod.swap` takes an ordered pair and swaps its members and thus does the reflection on the identity line. The function `λ(x,y). (1 - x, y)` inverts the orientation so that the resulting cube is type II (i.e. the resulting cube is type I w.r.t. `-j` and `i`), which is the right form for applying Green's theorem. We show that the resulting cubes are type II regions and that all of their boundaries are piecewise-smooth based on the following general result.

```
lemma typeI_to_typeII_works:
      assumes "typeI_twoCube (orient, C) i j"
      shows " typeI_twoCube (orient, typeI_to_typeII C) (-j) i"
```

Next we show that the type II parameterisations are divisions of the type I parameterisations: i.e. they represent the same domain in $\mathbb{R}^2$. We first show that inverting the orientation of a cube preserves its image as follows.

```
lemma rev_orient_same_img:
  assumes "typeI_twoCube (orient, C) i j"
  shows "cubeImage C = cubeImage (C o  (%(x,y). (1 - x, y)))"
```

Then we show that reflecting the diamond (resp. the disk) on the identity line preserves its image.

```
lemma diamond_swap_eq_img:
    "cubeImage (diamond_cube) = cubeImage (prod.swap o diamond_cube)"

lemma disk_swap_eq_img:
       "cubeImage (disk_cube) = cubeImage (prod.swap o disk_cube)"
```

Proving this depends on mapping every point in `unit_cube` to another point in `unit_cube` such that the type I parameterisation maps the first point to the same value to which the type II parameterisation maps the second point, and vice versa. Again, to prove this, there are no conceptual challenges, the only issue is that we need to manually identify bijections that do the aformentioned mapping for the disk and the diamond.

A last and a relatively challenging goal to prove is that the type I cube can be obtained by adding only vertical boundaries to the type II reparameterisation. For the diamond, formally this goal is shown in the next statement.

```
lemma diamond_cube_is_only_vertical_div_of_rot:
  shows "only_vertical_division (∂diamond_cube)
                                 (1, typeI_to_typeII diamond_cube)"
```

To prove this we show that some subset of the 1-chain that forms the boundary of the type I parameterisation has a common subdivision with *all* the horizontal boundaries of the type II parameterisation. Every 1-cube in the rest of the type I diamond's boundary has to either be i) a degenerate path (i.e. a path that maps $[0,1]$ to one point on the plane) or ii) a subpath of a 1-cube that is in the horizontal boundary of the type II diamond. As shown in Figure 2, the boundary of the type I diamond is made of two line segments and two degenerate paths. We show that the 1-chain constituting those two line segments has a common 1-chain subdivision with the horizontal boundary of the type II diamond, which is also made of two line segments, that are rotated nonetheless. That common subdivision is shown in Figure 2i-2l. Finally, the statement of Green's theorem for a diamond follows.

```
lemma GreenThm_diamond:
  assumes "analytically_valid (cubeImage (diamond_cube)) (F_i) j i"
    "analytically_valid (cubeImage (diamond_cube)) (F_j) i j"
  shows "^H∫ ∂diamond_cube F ↓{i, j}=
                 integral (cubeImage (diamond_cube))
                              (λx. ∂ F_j / ∂ i |x - ∂ F_i / ∂ j |x )"
```

For the disk, showing the existence of a common subdivision between the boundary of its type I parameterisation and the horizontal boundaries of the type II parameterisation is not possible. This is because the boundaries of the two rotated parameterisations have different *curve speeds*.[4] Accordingly, instead of showing the existence of a common subdivision, we show the existence of a common reparameterisation between the two boundaries. Since that common reparameterisation needs to be piecewise-smooth, we choose it to be an angular parameterisation since the *velocity* of the Cartesian parameterisation of the disk arcs has a singularity. The challenge in proving this is manually devising the mapping ϕ used to show the reparameterisation relation.

Another aspect of applying the theorem to the diamond and the disk that is worth mentioning is instantiating the vectors `i` and `j` on which the locale `green_typeI_typeII_chain` and accordingly Green's theorem are parameterised. We instantiate `i` and `j` to the concrete Cartesian basis `(1,0)` and `(0,1)`, respectively. The most challenging part of this instantiation is proving that the vectors used for instantiation satisfy the assumptions of the locale `i_j_orthonorm`. The assumptions of `i_j_orthonorm` are all straightforward to prove, except for the assumption stating that Fubini's theorem applies to the vectors `i` and `j`. We need to prove this twice, once for the vectors `(1,0)` and `(0,1)` (to satisfy the assumptions of the instantiation `T1` in the locale `green_typeI_typeII_chain`) and another for the vectors `-(0,1)` and `(1,0)` (to satisfy the assumptions of `T2`). The proof for the vectors `(1,0)` and `(0,1)` is, albeit being tedious, a straightforward application of Fubini's theorem. On the other hand, for the vectors `-(0,1)` and `(1,0)` the proof is trickier and it can be seen as proving Fubini's theorem on rotated basis vectors. The way we do this rotation is using the

[4] Rutter [13] explains curve speeds and velocities.

multivariate change of variables theorem. Overall, the proofs for this instantiation are 200 lines long, which is not too bad given that the instantiation using the Cartesian base vectors is quite reusable.

6 Conclusions

We formalised a statement of Green's theorem in a fairly general form. The theory and concepts we develop here can be used in proving more general statements [9, 10] of Green's theorem. For instance, proving Green's theorem for regions with rectifiable boundaries is done by approximating the line integral on the region's boundary by that on the boundary of a finite mesh of squares that approximates the region, where the accuracy of the approximation depends on the number of squares. The statement that we formalised can be used to reason about such finite meshes. An original aspect of our work is that we avoid the classical line integral cancellation argument, and instead resort to a homological argument. We do so by assuming that the division is done by inserting only vertical edges for the type I division, and only horizontal edges for the type II division. We conjecture that this added condition on the division represents no loss of generality.

The application examples show that our version of Green's theorem can be applied with reasonable effort. In particular we needed 348 lines of proof script for the diamond and 566 lines for the disk, including comments. This is relatively high, but typical of geometric statements, where "obvious" facts require much effort to prove formally. Nevertheless, this is modest compared to the 14K lines of proof script that constitute the entire project. However, we believe that applying this theorem to more practical examples can help in understanding what features are common between practically prevalent parameterisations which can help in building automation infrastructure, especially for finding subdivisions and/or reparameterisations between boundaries.

Isabelle Notation and Availability All blocks starting with isabelle keywords **lemma**, **definition**, **fun**, and **inductive** have been generated automatically using Isabelle/HOL's LaTeX pretty-printing utility. Sometimes we have edited them slightly to improve readability, but the full sources are available online[5].

References

1. Abdulaziz, M., Paulson, L.C.: An Isabelle/HOL formalisation of Green's theorem. In: Blanchette, J.C., Merz, S. (eds.) Interactive Theorem Proving — 7th International Conference, ITP 2016. pp. 3–19. Springer (2016)
2. Apostol, T.M.: Calculus, vol. 2. Wiley, 2nd edn. (1969)
3. Federer, H.: Geometric measure theory. Springer (2014)
4. Green, G.: An essay on the application of mathematical analysis to the theories of electricity and magnetism. (1828), online at `https://arxiv.org/abs/0807.0088`
5. Harrison, J.: Formalizing basic complex analysis. From Insight to Proof: Festschrift in Honour of Andrzej Trybulec. Studies in Logic, Grammar and Rhetoric 10(23), 151–165 (2007)

[5] `bitbucket.org/MohammadAbdulaziz/isabellegeometry/`

6. Harrison, J.: The HOL Light theory of Euclidean space. Journal of Automated Reasoning 50(2), 173–190 (Feb 2013)
7. Hölzl, J., Heller, A.: Three chapters of measure theory in Isabelle/HOL. In: Eekelen, M., Geuvers, H., Schmaltz, J., Wiedijk, F. (eds.) Interactive Theorem Proving — Second International Conference, pp. 135–151. Springer (2011)
8. Hölzl, J., Immler, F., Huffman, B.: Type classes and filters for mathematical analysis in Isabelle/HOL. In: Blazy, S., Paulin-Mohring, C., Pichardie, D. (eds.) Interactive Theorem Proving — 4th International Conference, pp. 279–294. Springer (2013)
9. Jurkat, W., Nonnenmacher, D.: The general form of Green's theorem. Proceedings of the American Mathematical Society 109(4), 1003–1009 (1990)
10. Michael, J.: An approximation to a rectifiable plane curve. Journal of the London Mathematical Society 1(1), 1–11 (1955)
11. Nipkow, T., Paulson, L.C., Wenzel, M.: Isabelle/HOL: a proof assistant for higher-order logic. Springer (2002), online at `http://isabelle.in.tum.de/dist/Isabelle/doc/tutorial.pdf`
12. Protter, M.H.: Basic elements of real analysis. Springer Science & Business Media (2006)
13. Rutter, J.W.: Geometry of curves. CRC press (2000)
14. Spivak, M.: A Comprehensive Introduction to Differential Geometry. Publish or Perish, Inc., University of Tokyo Press (1981)
15. Swartz, C.: Introduction to Gauge Integrals. World Scientific (2001)
16. Wenzel, M.: Type classes and overloading in higher-order logic. pp. 307–322
17. Zorich, V.A., Cooke, R.: Mathematical analysis II. Springer (2004)